Modified 5E3-Type Guitar Amplifier

By William Terry Newton (wtn90125@yahoo.com)
This document was last modified November 28, 2025

The Fender Tweed Deluxe instrument amplifiers were manufactured from 1948 to 1960 and included the 5A3, 5B3, 5C3, 5D3 and 5E3 (not counting the later non-tweed Deluxe and Deluxe Reverb amps). The circuitry of these varied, in particular the phase inverter design, but they all had two channels each with their own volume control, a tone control that affected both channels, and used two dual-triode preamp tubes (originally octal 6SC7/6SL7 then later 12AY7/12AX7), a pair of 6V6 power tubes and a 5Y3 rectifier tube to deliver about 15 watts into a 12" speaker. The most popular version was the 1955 5E3 with a "cathodyne" phase inverter and a particularly aggressive tone.

Kits for making a 5E3 replica are widely available, I got mine from Weber Speakers. The cabinet is built to order, I chose pretty much the stock options with a Weber 12A125A alnico speaker. Here are the contents of the kit...

The cabinet, speaker and chassis are custom-made in the USA but to keep the cost down some of the electronic components are imported. I don't care, was planning on using my filter caps and upgrading the other parts as needed. The kit came with a Weber WY3GT "Copper Cap" solid state rectifier but otherwise does not include tubes. I already had everything I needed but if buying new figure about $100-$150 over the cost of the kit for tubes and other parts.

I chose the 5E3 design because it was simple and lightweight enough to carry, but still had adequate output power for most situations. I had already planned on adding a master volume and an internal channel couple to avoid needing a patch cord to strap the channels, the rest of the mods came out of need and to experiment. The original preamp design didn't do so well under heavy overdrive with the master down, it had too much low-end gain causing "blocking" distortion, and the controls came on too fast making it hard to blend the channels. To fix those things I used a capacitor to couple the channels together so that a low-cut was applied to channel two, and added a "mod" switch to improve the taper and further control the lows by adding a load resistor to the output of the volume/tone controls.

Warning! Potentially lethal voltages! Do not attempt these mods unless you know how to safely work on tube amps.

The 5E3 design makes it fairly easy to add a master volume between the V2A gain stage and the V2B phase splitter. It can be done without adding a capacitor, just replace the 1M resistor with a 1M control pot with the wiper going to the grid, however wiring it this way causes the volume to increase rapidly as positive feedback from the phase inverter drastically increases the effective resistance of the bottom half of the control. Instead I wired the master volume control conventionally - added a 0.1uF capacitor from the plate of V2A to the hot terminal of the control, grounded the cold terminal, and connected the wiper to the existing 0.022uF phase inverter input capacitor. All I had handy was a 250K control, this reduces the maximum gain but that is not a problem with this amp and the lower impedance reduces noise pickup. Originally the master volume control was underneath the chassis using the extension speaker hole but moved it to the top panel where the fuse holder was, moving the fuse to an internal in-line fuse holder.

The 5E3 uses 0.1uF coupling caps between the first stage plates and the wipers of the volume controls (which is backwards from how volume controls are usually wired), as a result the volume controls "came on" rapidly, couldn't get the volumes much past 2 without overloading the preamp. Also, at higher settings the 0.1uF coupling capacitors were operating into a fairly high impedance causing excessive low frequency response and blocking distortion. To fix these issues I put a 1000pF capacitor between the normal channel's #1 jack and the switch contact of the bright channel's #1 jack (rather than grounding it), so that plugging into the normal channel with nothing plugged into the bright channel would put both channels in parallel and cutting the lows to the bright channel, about 6db down at 160hz and 6db per octave below that. To improve the control taper I added a switch (placed between the preamp tubes on the bottom of the chassis) that switches a 100K resistor from the grid of V2A to ground to load the output of the volume/tone control network. Besides making it much easier to get cleaner tones and mix the channels, this helps to further cut subsonic frequencies by limiting the maximum impedance seen by the 0.1uF coupling caps. Between the master volume and the mod switch I could now get useful overdrive and go back to the stock behavior by switching out the "load" mod.

This was close but there were still a couple of issues. When playing live with drums the amp would feel like it was losing power when taking leads, this was a result of the cathodyne phase inverter which tended to clip heavily only on one side with significant duty-cycle shift. To fix that I increased the values of the 6V6 grid stopper resistors so they wouldn't load down the phase inverter (especially on the cathode side), and decreased the value of the phase inverter's cathode resistor to provide a bit more headroom in that circuit. Lowering the master volume slightly also solved this problem but it was a fine line between lowering it enough to avoid the issue and actually lowering the volume, with this mod I could simply turn the master volume all the way up and the duty cycle would remain relatively balanced without losing power. I also ended up replacing the provided 6600CT:8 output transformer with a Hammond 1751M 8000CT:4/8/16 which provides better matching. At least with a resistive load, no idea what the impedance curve of the Weber speaker looks like so I make no claims as to whether or not it's a better match to the provided speaker.

Another sort of issue I had was at high gain settings with the volumes above 10 or so (on a 1-12 scale) it would still get too funky for my tastes so I did something possibly controversial - I added solid state components. The funky sound happens because when the 2nd stage V2A is overdriven the grid conducts on the positive peaks, negatively charging the coupling caps and driving the stage into starvation, shifting the duty cycle and producing a waveform that looked similar to tombstones in a graveyard - see the simulation section for preamp distortion in the stock position. It was a bit better in the load mod position but it still happened at high gain settings. The fix for this was a string of three 1N914 diodes and a 6.8K resistor in series, placed parallel with the 100K resistor for the load mod switch setting, with the diode cathodes pointing at the grid. The diodes conduct away the excess negative charge, eliminating blocking distortion. The resistor value was chosen to produce a waveform similar to the grid conduction side of the waveform, and the number of diodes was chosen so that there was minimal impact on the clean tone with the master volume all the way up. With this mod to the load mod I can now crank all the controls to 12 and the overdrive remains tight and usable with no blocking distortion, even with overdrive and delay pedals before the amp.

Eventually I added a negative feedback position to the mod switch to get cleaner tones, I implemented negative feedback by connecting the feedback resistor to the negative end of the 25uF bypass cap so it can be switched from stock to NF by grounding or lifting the connection. The remaining mods were mostly for safety and reliability with little effect on the tone - better filter and 6V6 bypass caps, bleeder resistors, screen grid resistors, protection diodes for the rectifier, etc. I'm very pleased with how this amp turned out.

Here's a schematic of my modified 5E3 amplifier...



This schematic does not show the impedance selector switch. On my amp I used a dual 6-position wafer-type rotary switch, wired so that both poles were in parallel. The first three positions were wired to the 4, 8 and 16 ohm output transformer taps with the remaining three positions all wired 8 ohms to avoid an open load.

Here is what the output looks like on the scope, power amp clipping and preamp clipping...



The diodes only act on the "cold" side of the waveform (the positive side of V2A's output but the bottom of the speaker output as the power section is inverting). The 6.8K resistor value was chosen to produce a waveform shape which was similar to the grid-limited "hot" side. I prefer the creamier sound with the diodes but it was pretty good without the diodes, just more touchy at high gain settings. If the negative feedback mod isn't used then the mod switch can be wired so that center off position is stock and the two outer positions are for load with and without the diodes.

Here are some pictures of the build...






2 AMP
OFF
STANDBY
FUSE
ON
TONE
VOLUME
VOLUME
BRIGHT
NORMAL

Simulating the Modified 5E3 Circuit

The LTspice program was used to simulate the modified 5E3 circuit...



The schematic is set up to simulate the input plugged into the normal channel #1 jack. The simulated potentiometers have a 20% audio taper and have a range of 0 to 10 for all the way down to all the way up. Several components are just for the simulation, these parts have designations like Rsource, Rextra1, Cavg etc. The simulation can be set up for transient response with signals plotted on a timeline similar to a scope, or AC analysis which plots the frequency response (but does not take into account distortion). The power supply can be a full power transformer and rectifier simulation or a fixed voltage supply which provides much faster results when doing stepped analysis.

LTspice simulations rely on models to describe the components, the program includes a huge number of models but for tubes, so these are supplied in the illegible text on the bottom left, plain text listings of the models are included in this document along with an encoded zip containing the simulation files. LTspice includes symbols for a triode and a tetrode, so these are used for the tubes (the 5Y3 model uses the triode symbol), I used the urc2 delay line symbol for the potentiometers.

Transient response with 150mV input at 420hz, volumes 10%, tone 70%, MV 100%, "load" setting...



Simulated output power is about 11.9 watts, slightly into clipping once the supply has sagged. The plots track several signals in the amp from the moment the standby is effectively flipped on (tube warmup is not simulated, the tubes are assumed to be warm when the simulation starts). The top red trace shows the main supply and ripple with the blue trace through it showing the average. The light purple trace under it shows the screen supply and the gold trace under that shows the preamp supply. The light blue and gray traces show the plates of V1a and V1b, the green trace is the plate of V2a, the dark purple trace is the plate of V2b (eventually driving 6V6 V3) and the dark red trace is the cathode of V2b (eventually driving 6V6 V4). The light green trace shows the 6V6 cathodes, and the blue trace at the bottom shows the speaker output. The first pane shows the full 10 second simulation, the second and third panes show details at the start and near the end of the tone burst, and the final panel shows the speaker output waveform.

The simulation was ran with my 12AX7SMP model which was derived from measurements of the JJ ECC83S installed as V2 in my amp, the simulated voltages closely match the measured voltages. This particular tube draws less current than specified for a 12AX7, so the plate voltages are a bit higher than normal.

Transient response with the volumes increased to 30%, rest of the settings the same...



This shows the output when into fairly heavy clipping, output power is about 13.8 watts. It still retains a bit of asymmetrical clipping. The downward slope on the clipped peaks is because of the limited low frequency of the output transformer, and is approximate as the transformer model does not take into account DC imbalance saturation effects. The specified inductance for the 1751M was way off (seems to be a theme with transformer makers) so estimated the inductance by using a low level sine wave with a series resistor. With a 122K series resistor and an input voltage of 2.367V at 422hz, I measured 1.336V across the resistor and 1.455V across the transformer primary. Capacitance was difficult to measure due to the inductance interfering with the readings but measured about 500pF from each outer and about 1100pF from the from the center tap to the secondary and frame. To estimate the inductance I made a mockup simulation of the test setup with two coupled inductors and by trial and error found values for primary inductance, parallel resistance and parallel capacitance (series resistance was directly measured) found values that closely replicated my voltage readings and was in the ballpark for capacitance. The figures that matched the best for each leg were 11.5H, Rpar=153K, Cpar=1050p (Rser is 200 for one leg and 213 for the other), for a total primary inductance of 11.5*4 = 46H. Another per-leg match with somewhat wider frequency response is 12H, 153K, 800p, or 48H total primary inductance. The 11.5H figures produced plots similar to that observed on the scope and the frequency response was similar to the specifications so went with that.

Preamp distortion with 150mV/420hz input, volumes 5, tone 7, MV 5...





With the volumes both increased to 90%...





This is very similar to what is observed on the oscilloscope, eases into distortion and the clipped waveform is nice and smooth. As pushed further the bottom side of the waveform (top "cold" side for the V2aP green trace) flattens out some, but not the harsh one-sided clipping produced without the diode/resistor network in place. The resistor in series with the diode is 6.98K in the simulation, which was the measured value of the old 6.8K carbon comp resistor I used, selected so that at most settings the diode side resembled the grid side. The source impedance of the driving signal determines the shape of both the grid conduction and conduction through the diodes, but for this design for most control settings the impedance is about 70K-80K, capped by the 100K load resistor and dropping only as the volumes approach their maximum settings which causes the conduction curve to become flatter.

Same control settings with the mod switch in the "stock" position...



The preamp distortion in the load position but without the diode/resistor network connected was similar to this.

Stepped response with normal volume 7, bright volume 9, tone 7, MV 5, "load" position, with and without diodes...



The input signal was stepped from 35mV to 600 mV in 8 steps. It still flattens out a bit with the diodes but the response is much smoother. I'm on the fence a bit as to adding a switch to enable or disable the diodes, as can be seen in this stepped response plot they do affect the cold side of the waveform before it would naturally clip. With the master all the way up there is only a very slight effect on the clean. When overdriving the preamp it definitely sounds better with the diode/resistor network in place, so for now it will stay like it is. At least until I can check out the negative feedback mod in a live situation, without the negative feedback option the mod switch could be wired to be 100K load, stock and 100K load with diodes. Diodes are temperature sensitive - forward voltage decreases with increasing temperature - so they have more effect on the clean when they are hot. The simulations were ran at 60C, the estimated internal temperature inside the amp where the diodes are. On my amp I measured 50C with 29C ambient, would have to be a really hot day to reach 60C inside the amp.

Simulation Models

These are listings for the SPICE models used in the simulation.

1N4007SM - a simple diode model as older versions of LTspice do not include a 1N4007.

```
.MODEL 1N4007SM D (IS=2.55N RS=42M N=1.75 BV=1000 IBV=10U
+ CJO=42.4P VJ=.75 M=.333 TT=5.04U)
*  600 Volt  1 Amp  3.5 us  Si Diode  04-01-2004
```

(the comment at the end is wrong, manually edited a 600V 1A rectifier model for BV=1000)

AUD20 - a 20% audio taper potentiometer model I hacked together from bits of other potentiometers. Use the "urc2" symbol then rotate twice to get the normal left to right flow with going up increasing the setting. To use right-click the placed urc2 symbol and change the prefix field to X, InstName to component designation (P4 etc), SpiceModel to aud20, Value to r=250K or whatever value is needed, and p=5 or whatever position 0 to 10 is desired. Make these visible on the schematic to show the values and allow direct editing.

```
.subckt aud20 1 2 3 ; modified for 20% audio taper, p=0-10 (fixed)
.params r p x=(25**limit(.000001,p/10,.999999)/25)-((1/25)*(1-(p/10)))
Ra 1 3 {x*r}
Rb 3 2 {(1-x)*r}
.ends
```

12AX7SMP = This is a 12AX7 model I made using the Model Paint program, adjusted to line up with the measurements I took from section A of the actual JJ ECC83S tube installed as V2 in my amp. This tube is likely aged as the plate current for a typical grid voltage is about 25% less than typical 12AX7 specifications, however this model closely matches the actual voltage readings from my amp. When it was new it was closer to normal specs, thought I had mis-measured but after reexamining the old voltage measurements it appears that it really did age.

```
**** TEST ** Advanced Grid Current ****
* Created on 09/20/2025 13:04 using paint_kit.jar 3.1
* www.dmitrynizh.com/tubeparams_image.htm
* Parameters chosen to model the actual V2A JJ ECC83S in my amp
* which may or may not be bad (less current than published curves)
* This version has improved grid parms and slightly tweaked main parms
*-------------------------------------------------------------------------------
.SUBCKT 12AX7SMP 1 2 3 ; Plate Grid Cathode
+ PARAMS: CCG=3P  CGP=1.4P CCP=1.9P
+ MU=106.83 KG1=1458.78 KP=632.37 KVB=513.3 VCT=0.3511 EX=1.523
+ VGOFF=-0.5786 IGA=1.563E-7 IGB=0.5073 IGC=76.03 IGEX=2.828
*-------------------------------------------------------------------------------
E1 7 0 VALUE={V(1,3)/KP*LOG(1+EXP(KP*(1/MU+(VCT+V(2,3))/
+ SQRT(KVB+V(1,3)*V(1,3))))))}
RE1 7 0 1G  ; TO AVOID FLOATING NODES
G1 1 3 VALUE={(PWR(V(7),EX)+PWRS(V(7),EX))/KG1}
RCP 1 3 1G   ; TO AVOID FLOATING NODES
C1 2 3 {CCG} ; CATHODE-GRID
C2 2 1 {CGP} ; GRID=PLATE
C3 1 3 {CCP} ; CATHODE-PLATE
RE2 2 0 1G
EGC 8 0 VALUE={V(2,3)-VGOFF} ; POSITIVE GRID THRESHOLD
GG 2 3 VALUE={(IGA+IGB/(IGC+V(1,3)))*(MU/KG1)*
+ (PWR(V(8),IGEX)+PWRS(V(8),IGEX))}
.ENDS
```

12AX7JDA - This is another 12AX7 model I made with Model Paint, but using parameters from an ElectronicDesign article by "Jack" Dyer, title is in the comments. I reused the grid parameters from the 12AX7SMP models as they seemed to work well. This model lines up closely with published 12AX7 specs.

```
**** TEST ** Advanced Grid Current ****
* Created on 09/20/2025 13:38 using paint_kit.jar 3.1
* www.dmitrynizh.com/tubeparams_image.htm
* Parameters taken from "Modeling on Mondays: Nonlinear
* SPICE Models of Vacuum-Tube Triodes (Part 3)"
* This version has improved grid parms
*-------------------------------------------------------------------------------
.SUBCKT 12AX7JDA 1 2 3 ; Plate Grid Cathode
+ PARAMS: CCG=3P  CGP=1.4P CCP=1.9P
+ MU=102 KG1=1770 KP=766 KVB=78.75 VCT=0.588 EX=1.399
+ VGOFF=-0.5786 IGA=1.563E-7 IGB=0.5073 IGC=76.03 IGEX=2.828
*-------------------------------------------------------------------------------
E1 7 0 VALUE={V(1,3)/KP*LOG(1+EXP(KP*(1/MU+(VCT+V(2,3))/
+ SQRT(KVB+V(1,3)*V(1,3)))))}
RE1 7 0 1G  ; TO AVOID FLOATING NODES
G1 1 3 VALUE={(PWR(V(7),EX)+PWRS(V(7),EX))/KG1}
RCP 1 3 1G   ; TO AVOID FLOATING NODES
C1 2 3 {CCG} ; CATHODE-GRID
C2 2 1 {CGP} ; GRID=PLATE
C3 1 3 {CCP} ; CATHODE-PLATE
RE2 2 0 1G
EGC 8 0 VALUE={V(2,3)-VGOFF} ; POSITIVE GRID THRESHOLD
GG 2 3 VALUE={(IGA+IGB/(IGC+V(1,3)))*(MU/KG1)*
+ (PWR(V(8),IGEX)+PWRS(V(8),IGEX))}
.ENDS
```

6V6GT - This model is from Duncan Amps, seems to work fairly well although as noted in the comments the grid current is approximated. But it seems pretty close. Screen grid current at idle was a bit lower than what I was seeing with 6V6EH tubes so for the simulation added a bit of extra resistance to ground to compensate.

```
* Simulator:      PSpice
* Device type:    Power beam tetrode
* Device model:   6V6GT
* Author:         Duncan Munro
* Date:           4/5/97
* Copyright:      (C)1997-2000 Duncan Amplification
* V2 [12/10/97]: Screen current limited to prevent screen current
*   draw at Vs = 0.
* The following parameters are not modelled:
*   (1) Heater
*   (2) Grid current is an approximation
* Please note that this model is provided "as is" and
* no warranty is provided in respect of its suitability
* for any application.
* This model is provided for educational and non-profit use.
* Email queries to postmaster@duncanamps.com
* [6/19/25] WTN - replaced spurious {} with ()
*-------------------------------------------------------------------
 .SUBCKT 6V6GT A S G K
* Calculate contribution to cathode current
Eat    at   0   VALUE={0.636*ATAN(V(A,K)/10)}
Eme    me   0   VALUE={PWR(LIMIT(V(A,K),0,2000),1.5)/1300}
Egs    gs   0   VALUE={LIMIT(V(A,K)/600+V(S,K)/14+V(G,K)*0.65,0,1E6)}
Egs2   gs2  0   VALUE={PWRS(V(gs),1.5)*1.45E-3}
Ecath  cc   0   VALUE={LIMIT(V(gs2)*V(at),0,V(me))}
* Calculate anode current
Ga     A    K   VALUE={V(cc)}
* Calculate screen current
Escrn  sc   0   VALUE={0.7*V(gs2)*(1.1-V(at))}
Gs     S    K   VALUE={V(sc)*LIMIT(V(S,K),0,10)/10}
* Grid current
Gg     G    K   VALUE={PWR(LIMIT(V(G,K)+1,0,1E6),1.5)*(1.25-V(at))*650E-6}
* Capacitances
Cg1    G    K   7.5p
Cak    A    K   9p
Cg1a   G    A   0.7p
.ENDS
```

5Y3GT - This is another model from Duncan Amps, it was slightly modified to use a triode symbol.

```
*--------------------------------------------------------------------
* Filename:       5y3gt.inc  17/5/97
* Simulator:      PSpice
* Device type:    Valve rectifier
* Device model:   5Y3GT
* Author:         Duncan Munro
* Date:           25/4/97
* Copyright:      (C)1997-2000 Duncan Amplification
* The following parameters are not modelled:
*   (1) Heater
*   (2) Reverse voltage breakdown
*   (3) Saturation
*   (4) Interelectrode capacitance
* Please note that this model is provided "as is" and
* no warranty is provided in respect of its suitability
* for any application.
* This model is provided for educational and non-profit use.
* Email queries to postmaster@duncanamps.com
*--------------------------------------------------------------------
.SUBCKT 5Y3GT A G K  ;modified to use triode symbol
GP  A  K  VALUE={2.69E-4*(PWR(V(A,K),1.5)+PWRS(V(A,K),1.5))/2}
.ENDS 5Y3GT
```

Finally, this subcircuit model is a simple 2-pole 4th order lowpass filter set to attenuate frequencies over 7Khz to roughly simulate a speaker so that when doing wavefile simulation it doesn't sound like mosquitoes in heat. It is designed to use the urc symbol rotated twice, set Prefix to X, InstName as needed, SpiceModel to RECFILT, and Value (for this simulation) to Rattn=800 and make visible. Rattn sets the initial attenuation resistor value operating into a 100 ohm input load and 100 ohm output load. Select so that the output voltage remains under 1V to avoid digital clipping.. For Rattn=800 the maximum input voltage is (800+50)/50 = 17 volts.

```
* 7khz recording filter
* params: Rattn - series resistance
* typ 500 to 1200, less = more output
* use something like URC symbol
.subckt RECFILT 0 in out
R1 in 1 {Rattn}
R2 1 0 100
L1 1 2 1.74m
C1 2 0 0.042u
L2 2 out 4.2m
C2 out 0 0.17u
R3 out 0 100
.ends
```

If I wanted to get fancy with it I could add a frequency parameter then scale the L and C values in proportion to frequency/7000 to make an adjustable filter, but 7Khz sounds about right to my ear. This model was made to reduce the clutter on the schematic, previously was just using drawn resistors inductors and capacitors. Similar techniques can be used to encapsulate other circuits needed for simulation. One such subcircuit I'd like to make is to combine the filter with a more accurate model of a speaker, in the present simulation the speaker is simply an 8 ohm resistor.

LTspice Simulation Files

The following is a base64-encoded version of a temp.txt file then use a base64 utility, typically using a command like base64 -d < temp.txt > temp.zip then unzip the temp.zip file to recover the simulation files.


```
UEsDBBQAAAAIAJxzQ1v/zI2WwxwAAKtfAAAOABwANUUzbW9kX1gxMi5hc2NVVAkAA/gj4Gj8I+BodXgLAAEE6AMAAAToAwAA5DxtU+NGk9/zKya5uopNFqHRu5Ki6lhgyWaBUOB1
8tTDU08JWwsKtuST5GX3Uvnv193zotFIJske347awLhnprun33skZ57XTVGVLHD4Vzc/np7OGGe+G0eMu2kYffXL2+tT5vsR2w8B5oViIMCxAqt5CU4CAY57YM5DT8LlAgn3Irle
r5ATgZvICbVETkR6R5AkvYnElRNRf0ecSto8jtwhcYBZNATvQTB+CC8ZMCso4ISiNZiI0ohGYkLIknuWUIUsAdwXKo9c4DsADDx0Axop/JGaUEvMowWJOlo3IRAAzEIVBbg/SBie
GweGKAIu9Qn7DDV3YC6RJKgCBCeAqAPzMBIfYYSC6yZIUzQRjWhZTPRRiRPzqDu7mBBsgxD6/AtNIVyO9IRHAusmXFdNoKZoQohOT/heoibkEmWrnqQtrNaYCBUqtcRQvw9CoUHg
e1L9qBENhoEEo/Twl5oXYFQigWG/AUbzEeDIXC14RrjFPW0GCbj0RxucAHLuGWCBEcAWahKgWN4XZerKDTyhE2n8qR/JCbXEMDc0/r5dye0AG0eEEwqlKWHPtURJBopgZamG8aB0
xcDXVkgyA7jlvsL2aEKM/GgwofYqU4hBMh5ELDHqTmBMoAvCSEzgOO6rRYL2Ufow6PQnF2oY6Q+BSpE2mHvGag+NVq/WYFIBgpVSTG0LLGHvJLFGr6TQsQ1aMHGTASFQDBJTPQhW
ejJt1wuU5+n15gSGCRiZ8YMmVCQxVSom+goiw6IJMVL5TxgWTYjRyITaK+WJvo6/OEZ/GEgjk59YQCwoMAVeBKsIbIHDHpIIc55ercGpJ8EpN1dz11XwtA8PI8miWqGU6yre1RIl
aZKvmwwkLaSAE0oeJm0UgU2DlEYTSn3KFfBMNNF3BZHcaIKSFoxMk6MJtcRMvDRBIyOSaK5oZEx4gZrwTIshCwVCcpB2voju4tKcmVuBd5tNcliOKup5buxKsBjEZhJFsBikpsuk
ig3uGWYC0L45KCiFs9SsatTStJeaU5WZ017+TaPOkQbwIDGWQzZl+1ht0KCTCEfsGp5of3bVcjFvGkaivFyf0YAj9aSXj/XypJeNE5WMk14uTlw9bx4IyyF5MrOaSCwBugKo9Cah
lPISNSt1SEULqJBHHTCVSOWkKbogsUSECQEMTKQvbpaHXJbcXdUgoXKLUTOopTrNYXAAIBUMCkhcAVByJ40tYRghmZYSWjrapNudz+0fJEkjCTLdGQs15dZGwiMw5cEBlDZ1Sdz3
+jIgPfme+Ot22pLK4oaxR4Gy+R4MOwFdx4Ly5DIFwhMAKPG9DkReI07Yh9BaI95FKqJ2xuvGngbrtahrBOJfLQIFFDu60IFhQFiphAFrEtaZMUJMhTGRQfUKtDtXwDudEsRlFgRD
jzJDN1GLDFLk/x0A+UaQ4N9Qfmrpk4d+pIGpmZEJYz8ZaDARlfwRe3iJyQ7qX3qHr2eXhUsEHx1mkWgqTUFi0Jzodglaz44QLYYwidXkdw088RiH876/A6oI4dkG2FvNW06Tle
0hcURakg6OtdJEur0qD8DVDPRKqAcouhEZGx/O5UtBQ7frFFAkVhpyZV8ekKoJjUFiJScgdS1VB/cyKAPW1St6SAKmaEkm/qpBWQVJZaqtNAz9hOfQBYsuS2Ux0Wb/TXNaMxlXTc
XEt6IDAN9GoUGoFpoMFoIaLENGWAeqNQFpFmOsFg4Ka/sclvIvg1anqwRt1ZdDcAsQE1qlkgL/qT2AQj+9Qt+eLmRp0jFFcOVLcYzYy83ukjQf4pvUTmWg/CBHWXmNK71jCSi6le
MDpG7E9d116NSESDLbCpRB+LnCUGOppgyujAOozi0QlMA72aFqG391cTkiDR2Ay+CSwPoCTCvQ7c3SXECnfcY7ADi20mEoyqfTCKAjO0EE5XFMFKApNwhmC5zUTiJTbY59J5fHG5
ZFRQAgy/DDC4mwAHporReAQYztX1wmg8BEbj6cAoffJbGnBTsHTl1pdgB8ZEHrmGqMTNHQ46MKyMsOWIgW6k/S5JEgFGkh0YrxgEODFXo4QoTNBAgwllMsBNJDU46ZFMFMmkRzJJ
9LxSGtYaGFiwAI49U5cCLOaVvIFnAvsQ0AxwIFfLeUMNBA49EzeGIAEOTSSR4kTOG4cncP+UxC4Ggj7fhNKPbNx0CrpoAi+Me84twGLewI1lshi4Ju5E3Rgi+M350ZmOPfKjasLE
J5l35CcVWeVHFVPlR33jY3wO9KzqteRHlSoUHZUk1GfdGyuAShfqs8oTmlFRw+r96rJWAXRPrTdwC4G6xWi2myMDqYK9loyr23gAHUuxyMZizrMrKRl16zjndxKkr+HmnlrVXfLN
vTsNk+0dgN4JkL5guwp+lIv0XQOsOpPSVlchgP9MnVndFVbb1sAlmo5OrNSi+8E7BVJXr3P/zAQhU8GZwSZ6usYjY776rKxefVZxpBOkNkeNQvqmKX/padnHe8m/LCPVHlUC6gOq
kmrOHztDoxODXN6ZzCSJoULVUoA8FXv6AcHc35ggFNY8UCAVkvW5ZXRVziX6iqJUZik90zRTPKDmX3t6nS8QePOPi9c/n7OP1arN7nNR6ADJa/Tpy5Off8GIwEgIATvPP7TMUxN+
yrDDJKBeDekuhg5NLQXsR7PZNZtnq23usaNjxjXsbdm0l9k6B1H217Gbt5enE6zZ3DULgPmQiVY2cqeK3zpv5D1CYvBKURr0x9l1cf9g8grAFIKyrycGXFw31bZe5BYr3H00SVKz
BfSGu+0z8Iv83tzJqbFg16nBLJwMtP+6attqbfLqezQxqzajjFpiBUt7tClhtHgJUv6fkIpEML/miUELohDUFLa1QFCDkjmyLaOjFfyZBMVFLpw/Noj1DqBpWccdEgv//GAYS1+E
VrSb1iLbkBnvSzsebD4e2JXruhu1/aJoFre3bV1Uy1z0KRieOn9AoYcYeyxluARPvFE3BR86+jW+ubga89S7naQT1yKMtuGPUfbRaL6AdmZqiG6vIQ6OOmNsYU0815S4L9rHcYHb
HHnh1qSLeZKeDo8STOba6tkVhtbdu9NndiPf4qGhZ3s2QMY9Gyd2ePax7dmuw7dDWskgYn0RMdu1TWLCkCD0yo4kYhdGNMHsA1nf40M7ClJIjDsSzeZwJM9c2b5UH/K1iDEEvKrz
D8Un9qsG3GyKRX4BBr5i2XbZ2ZDFMkUKm2nIjmNMhzEUUn+LadseX4DpQD6k6MVtJeloXNK7UvrmMB5h2rauv8H00fuTvrdi1UdF5qi72oE8NMKj3r0rUBzbodl1XDeILdUuUQ48
W4UyaAwTFH/PRDLl/tRe8vGTXHPXxuCEPfenBxPDGuKLPNKOkFaUiynKRbtqnZGk1ONUPLnEMvIFWLWjIujI88bCR6KeUQ7MGqpSux7BR28pSwZFLda68TNhZUQeV3Z4qw+9UAjk
77koSl6IbkdyuuaDfGkUST1rlY/8d5rr3V8wV/FkORpHcc1tJ+ePg807q2VbZGFkbVavc41vH3h91LM/2Z0Pi+AvMsCBY9q5UpBLX4ia7VwmNZKNR/16sks2dkzzvH4NIvbvNrFB
aDD2SxPL25psTLyOxseD63zQPMyjs9lzmNIdiGxj6SESLY83ru8vanq4XcX5/uOQ2lDdX0bMDm9BPCS2M2d4tmlC4OlZZkiXjuOmcjyIJlYWkO/hDCvOL2tbbcMOYndIjY76EtTs
sxnUhM9iYbArxR0PgpuRcmg7vUshQqNRKAVQfwwbX3wrdVA/GbRs+zZokWDIOfD+rSeYKGSRPyYZumh/RjLDo/UtDh+BDMl9qR7swwVOZ+BFucTbtEBZmSlLD594DWXJXSyt5zuv
cc4HtQmUURo0+7zJkWo/FZ8XJd7/5JC3geT1JoNyNfQf2TGN3LCrKA2G8VK2zDMDPmFijp5jt1BQv9b7LruX2A3iVU11r+mAQkH4YDjEOxh9y3Z+bBvdIPoLzPsOrEwP21sqefq
6t5wJA7/DXlDaQ5qs87WVlW2tNhL+k5Ljy/7tLA4TaEuAp/F/BmXtQsQKzyIt2xeqFW305FZEAgNY/0hCwJDhnS2YTCC6DC4fOn0O7iFc3can5OM6r1jC5+BkRCs60GIGCOeDVrw
n2HMlnjqOsEu1mLvL3BG1aXlEQmgHeUsSnZzZhdcz3I27ivmFbx4vm5rM8ZKfCS+QGLZH9yyYdMzvJvvX9iP9ATWOeQ1PORlBy+v+5fv1F+Eu+qSVZ492oZ0Nt6noE8Or1r3Y3py
1UXO7sQpvaf8TEydh0d2e/APf1B5GuTpQq5PPooh1SRj5Oll3uhZ8q+fI78UdOmtDgx7PbohWNrgghnc1Yufu2E+GeS8y8B1oZsboUlW3z+rzyEMD6hiewzFzHNkB7nLIkt2gu9Z
v1QxkX9q62xQvvCwX7/gk1t8KvcSFPHB4O6OW5TY8qsPo3XkcD8kTKvKjkfDJGK17+J8hHZBfYeAbGvwI6eXczE5wb+eePbxpfVR+eD7Ic+Ud3ZYtm+86S13f1f7YodOP30cGE8U
v6Dx2OnNtx8y4YvkeMc/yi+hsK0+6VouHcLla0zWY9S/H5Jt6QZe0uM2ki+X9woafOV+WNCErEsTI5qwu/4o6vQowof4rpzV9dL7ZKNNiP9cTXMyMJvLVLRBHT36wtuAHkbNUXr4
4uhueoOyxqRHV7OBO6QWxoh3hFoIXpE84xV2yRY5aWKJk153GySeXiloPGLsFY7D4w2u4szj6StSoIfvAKNdYqgZoHlvR47rrG3LQ3yTR8Gfu8u8Pj1+8/Z89tXs9NcZvqAh3/QR
Nse+fgfVC/yDEs9nXCxCdvCCXtg/+9pptneLx1bcikI3BcJgP7B1tSw+FPmSfahq5rn/idNFxdpsk9ev2ObQxaZwAmzly+lt6UAXlK0bVrMN+3Q48cK9vVWxLtqJ4+IPf7U54O4r
J6Wf6YEXTvcnE45/9yZ8f4Kz0ynguc7wi7Xs90979R/w6Q6vhtnvsOTTlCBOXi4bcQx68bE7qANBosR3EdC1PXcrFokvlAAHYl3AfrhQ5wpPfXZc1Itt0bKbYr1dZS1+w/epaB/Y
bHuHgWfR4tKasV9ml4D4wAeGBV6sFKFUUW0R+2FWlbmYEt+IiD099bqmomVereQCrDS76cuqXmerbjqR91xq/iJrWuCh2+6KtkOpb09xX9XfM/q5IvO4LffYSf4RRqyFopfmrqon
QHWXZ2smLwGNVWu0J1xGN304cbRtHzRW+DnZlgsQ8sW2rCvamLV5N8tYcBAepDHOHFebz3RsOT05nvI0jfc9VIVEc7TerEC+C5I7bpp77J/cA1MAJP/6nt0s6jwv2WJb13nZMrIn
0FtbsU2df0RQ01uBKBhb1tkTy0DcDTuEtI/A2UMOVrxaVU9Fec/IUnMQacOyOmdl1YqTr/Ll9wLFhE/ZjzmcrZafvSk7q4ulZqWArSXLNpu6+lSsNf9XUIA3hBFE/gA8tA+wkpDj
Flj9sVjCCb7JGvj8DeBY4rayYk9ZDdbbfu4tK0qMkRuwQlZ9YEXbsAZsNbsrVkX7GTeiZ2blZ2RkJcUozztKFpfny61YCBYH5IF2uQ8LPoAPbJucdp+us2LF/nub1wVEaJR21bRr
MsL/WpLmsvWmcRbVGlf/MzrgKbjFv8hH9oHjzSpbALVms4Vif9uw3/8QPjWZMtiw/zI/tyVzbt6/Pn43E9bKjtgNO2PvyPay1QL9IWeLqoSO425Ljg0ngaM/YBbQBnMKSoIf+u3C
f/Oj8/enh7+7TuRHe0ezo8vJfHL06t0UwxNEn1MI1/BDv43lV79cT87fXrydydWv3Fdo59NX3Alhq++6uPe+wb3029hr7juAtui7+eSGCAYwOoPRHjATAkZ+Gk0FGo/QeDYLN4Dl
vhFE97gThKf7Pm7AQzO2WIzTBUTTvfkka5Hr+WSdT5GKKcSs7InsLCOHPsJf7zp888liYe+03fMUACUDsC3teE8yMuEO3yduENcZSQwUa5NqFtM9dYAbIXBQECiJGDBdFZDcE5Iz
C4mpMxTzd1zKWMgPGPFCycleFLqn+5E83CZbgBOWi7y5LY/veQ937IQbgGaPPQmlG1qZqZU4AWcGqHN6eXJjJGUjK9ADeXC9Fg5RVBSvjPCvL0bYD+KxjMBBL7upiZ+4huKLaRrs
yTQjLlM12NdgE/tPgUTiivffFZweXP50ciQ3URPQVRQv5+V77E2xyiHi6DwTfvbvW6cowYh4rPPNX0+BUGN9zPE9TJHbx/If3Rx8Sf7zwoPg/5IAXyJRXUNqrCELqa7ors6zx2X1
VMol/pTdZO221kQBFkyhHAUs+QrEQo8GF52d/z/NbC9mwipPkVWB859hUPhBl9fABvDH5M1Y83l9V60gal1RnHin45XnROnpfrA3wcAl0wxGqu9k5NcAKK7/kJFFXoF1Lhp28UVw
M0HJ9UhPZSNBr1jhTaPacLfamlMYEPRUXfbaD09X5T/cCFlfOw5o4JzTo57qYQ2x89yjBynygy8eUqiP/AA6lmzVVOwh+wgWSM9XyCHA+lfs+LZEFdPzFwA67FpW+/jFnzDqOpuL
n09Oz/UNGThk7c3h54Thpfs+uYw8C7Y5SF34pC9nh9Sl/CWBu9vy+/Y8U8/wwonuGLznw5xzcWh40PbPZsdho4bvJ+S8+BNJpTmUEBw9GfGfCcEhULWKtgJyZW5wb7L0fEDg8ug
k5G+wBMyxO+C+J3gv34HQ4jqITuPVAdH3+M2tXke4IMBlB4ILzyE5kMMo0NormGo1IP/Hx0jF4irFeluy9+2TQtx57bEj3d5i11HtoBEmi0+v7otMf6AsMllF6Jvkmjpa4RBZw90
EU1IG91XyaV4sNhom94gZQhYGKmaLXj785TEQ1LDKqFLdo9n3wcHyQGPbsur/RmPQ35xW05Eda7MWXxlRm+DTmODPMogSdbJue/E4G7H0DdAB+3EKcJix9WwEIoqAevW4S09whIn
0LDYCXyEpaASBUudgAPM446vYDG4K+71vA5fDOajErQrvk6s+8BqyW6gkl48ELtXVQMmd9NWi0couxvIUy07EmCPXb4BGPgKqzZQgE0WOUb3qZj12XmVLbs9FC1fS5r0/f6uCOmg
hnpfG611ahghBWdsjQy1owpbyGqLVdY0xYKa7nuM9aB28BapXdEgZOxD/oTRvaED7uHLA2i8qo68h7quoYAA1gmfs+VSxE8kAGXQtw0zLBeSWWFklKsacteCWKKboJ2IdGXAGhBt
3iI8+1gV2NFk9SM5CFUlSATSJBy5BP4gPUIeqqs1yG/7Bjd5HvxFhibNCguB1eep4k1YOvtASCRGeQ2weMjKEkLcf3D2WwaqBXlCF1MCUx2Ld+JCAZfqdQ1ZBrRFDJbU1fYe5Umv
rb+Bg4ik/QDbgPqHOoccWC4wMKOHSYocCG03K0hFFaFRcI89PWAVrz8WTfktJVXK8UckvIyJPIoetV2jXNsnFO3cy8i+5t6dUDJFHEEJaxwQoInk5OpkxoS1Amsf2H0GpQJYhDye
6P0p6bMGLf8VK/N7MDTMEcDyhzxf3qE0kOQGlAImho+nQQXFuliB0QECqS6kDvLblngQyfZU3ELc6UYRqpQCCNVsmfd17ISPiIs/mib3kGfLuhJ9MdgjGaxGMbQTicP3Bb9B3EO2
WBWbjbQNJR6y3APNVAlCrmpCUqxRLjmD/+plDfJgk6oEEwPxkQCwZpoaSdq4X5pQVbfPIMEfQCYH9d5ieUZVpGD1FnXQQDhu0Xx+nn3bTLumBa+7dN0PP2x2ejNje8j0Rywfl6If
O5ZXJ7iE6uMa69YlA3900wPPhX9eCPnhez8B0xLlb1G2/34sWuc3UJzvcNz29PTkLNdFW38ui/95wFLtoN3e5eL28d/FGoP5Q0sauOrK5zZ7BFskqX9Dt6aIHyhfVOXyf9u5tuZE
kSj8vr+iN08akXARIZniwWGIcYwjq4TkIVVbqOxqRUxKNCl3Kv99v3OamzupvVRl52l8EWj6dmi+/s7XzYkP2QVevg2uJTHz6Gkw8HzB9/ELEsXz/T5tsyYYbiV4NAKan8zmSclK
n/MIgUuiwPJpLOQIQOOI1LwXo3yTXBb+WCnfBuyBS9PLwUycJuhNeqPphfC8vmuCYnr9APwHFMfz6OA8EHTX6AYcyBDDvu5ixtXEMACN6Iph9NG1HWJCkReCr1lwH/07ZDPPzzlf
1B9fXrptpNhOVwz6PaRZXdNv2zj5SDk028Shh+JUjY6QHbwFfPJ/Mo+vC1otlXCgK2bzbBicXo/7Db3l3wUNnDT0s9FNq4FOtaKGgTuaZ9Sb6S+TsIE+t2S2U/lHv1c0d5IXrcOl
/yDCsehF48EncXk97oWDL33xBcxzCh6vs56dNyAn73ZT8e9K5i7P0Ky+zgV7AWehgv+uZE8+6a94lK+4z+uFV0ho9ye4F6kgeijmKx4wpdJVN7juhT6lmVJjxzOv5yySJz7l5Z7B
gH1POHULkn3a/KApbzCeDsJB5HMFIrya+NOr8fUn9LvPrSv6jbHQwhA4w4HXKgx52hjdUK+bp2Tu3DZOU6FhUVqnOG++HsklxKHfG3m0zvsjz3z5mCWsPQuplvabveL7bk+Bv9MTn
z8L3PMecElCnB+JEXNdyhckujQ8CKE9/xIiBPjOaytZJlpVK9I44wNN+tl5ly4RVr+cka/5bUOJ5p2AnAjM2cBJzBU263w20pqPg/UCrqzqmxK2OBaRyCLq6pqGaNqOXBYpv5uhl
Wrou4cuC3/kDvn7A13eFrzfkVPoORPpbHHfuyLNK6Fvypz1e+G28yUATUzCxUgirVqISBSCRSXRL4M6nDHXPtK6cwacagX/ut0kK6MjwnoPhwi1ghzYSRIAu+DN917TOVUvhr/pd
UyuOPdcwbdVRKKiAC+e2Y8JXp+gIro77cZ3CIrjkOht0jHtUwzYVDo2Ay+AJfEzX4TcrHB7B1R141UrhIwxdCy8rGko9LuFB9vMFGLZIiNsuxOwgUu4KwXUNUys8JWxmOk0eCIPr
Pal7XNGSFEfkIkVzn84Ip3+LV1tA4HwNuMb/jCxNRmfGm0nHMSOP+QVYzI7ac7xdSWGSofZ+I1tJy4+Z+BajqbQKp9FOCdWiEf0u2jrtNo94vfI8jRm1aZ+JaqRxMzdGAdrFxPKS
bJPSHMxs08ImuLCId9Lvkh2oKqaEunmJMsqG76nKWgVcJpt2u1vN4Rb+I3P+T7yZW1eYG6N0vqTqy3by45MmQnPLCf2vVpg/Pq2K/u++GTJ5R+UCIdUnRc9UtpA0dc4o1XhFJAte
X8Yw28CPTBZq7i+RoliXt26k7yhuEwwdcVtprLrD0UqqOy9Lj8s5uh7meqXe5bgvlcbjSVdfhBMeu2yF2lJFkSx1Bd57UUhY9xiWFOkh3sTrA/zDomazewQ0jdma+rYtfUEOz92p
BDnjtui1zQG5tTdSLA7hVUnJRYKVR1IrE7So1s+jLfqZZDmd+m6IWoqMvlxT7ao0GQOxSjnJ0kcM9RN6thWGmnYlBZ8K+2H5B8ff2C7o+UnxhZiSpHAXct8N6WJSs5ZudrECsjs8
0Veu7PQbmqbIV9xFhdsCm+k2kvIzWq5bUh3r1UMibiZeKesXW23yTTuCYohQbkxbOh1iyuNW0N4XmgBpO4tGejhTJF21O2k+dWr8DYCxJyFdcCgY0QFe8NRJJ5Su20iemPk5l1Tf
P9M52kDz4XgjkebIKE5F6pvWy1ckKVJWbaCoL6SsWb6Jh/HrnW7QueA9R6o+FFFDxkBpVpmNmhoRPua1ABQeKP6AoNJQXeKuNjAyFSZz2kcZf1Z3CeC9q/30J1BLaWQUAwAACADO
mUFbrXiGb+oAAADfAQAADgAAADVFM21vZF9YMTIucGx0bdHNisIwFAXgfZ/i4kZl6nDzU027m9VshtnMIIp0EbVIsbaliYJI392mTRMXNiWQLycHQnb/jSxVnpUavkpZ3FWu0uAR
AMBvLctMJUDMwoj5dCMPPRJ4RJTTmIYYTtYzda3lfNJa5A73HoXDg8flgDcqa4/xiHuHDNHh2SeZRfJ6PBpxPC5iwWyyumoftJWMn12OrgaTt1Nn9tKbBGbTyzREMz5RdJOgUUyX
SLhgEdLV3Ca3O0xNGEx4wTHsf/TbJHVdJPtgfRd282JYjcF1WjlivBt2U91SgDNsPDd5Mc/fS8y+2Zt0AZPUEsBAh4DFAAAAAgAnHNDW//MjZbDHAAAq18AAA4AGAAAAAAAAQAA
ALSBAAAAADVFM21vZF9YMTIuYXNjVVQFAAP4I+BodXgLAAEE6AMAAATOAwAAUEsBAj8DFAMAAAgAzplBW614hm/qAAAA3wEAAA4AJAAAAAAAAAAggLSBCx0AADVFM21vZF9YMTIu
cGx0CgAgAAAAAAABABgAAApIhDEz3AGAI/MGZULcAYDkrU1CNNwBUEsFBgAAAAACAAIAtAAAACEeAAAAAA==
```